

# GUI 3 Widget\_Table 로 영상 표출




## 이 문서에 대하여

IDL GUI 프로그래밍에서 가장 많이 사용되는 컴포넌트 Widget\_Table을 소개합니다. 이를 이용하여, 이상우 박사의 IDL 블로그 중 "영상의 화소값을 도식적으로 표출해보자~" (<http://blog.daum.net/swrush/150>)에서 다른 내용을 다른 방법으로 접근해 보겠습니다.

## 가장 간단한 사용예

한 장 강의 'GUI 1'의 내용을 잠깐 보신다면 다음과 같은 WIDGET\_TABLE() 함수의 사용례를 보고 그리 어렵지 않게 사용법을 파악할 수 있습니다.

```
data=dist(10)
tlb=widget_base()
table=widget_table(tlb, value=data)
widget_control, tlb, /realize
```



온라인 도움말은 언제나 필요하지요. WIDGET\_TABLE() 이 가지는 수많은 키워드들이 아주 쉽게 설명되어 있다고 생각하지는 않지만, Excel과 같은 스프레드시트 소프트웨어를 자주 접한 덕분에, 각 키워드가 무슨 일을 하는지 파악하는 것이 어렵지 않습니다.

## 몇 가지 키워드(속성) 다루기

GUI 컴포넌트의 속성은 컴포넌트를 처음 정의할 때 키워드로 설정할 수 있고, 이후에는 WIDGET\_CONTROL 프로시저를 이용하여 변경이 가능합니다. 현재 상태에서 테이블의 셀 값을 변경할 수가 없지요? EDITABLE 속성이 활성화 되어 있지 않아서 그렇습니다. 다음과 같이 변경합니다.

```
widget_control, table, /EDITABLE
```


이제 각 셀의 값을 수정할 수 있을 것입니다. 물론 현재 상태로는 셀 값을 abc 와 같은 문자열로 바꿀 수는 없어요. 왜냐하면 모든 셀들은 하나의 배열 값들이기 때문인데, 이 예제에서는 처음 배열이 실수형 배열(dist 함수로 생성된)이었지요. 그래서 모든 값은 실수형입니다. 만일 문자열형이 필요하다면 처음부터 문자열형으로 바꾸어 넣어주면 됩니다.

WIDGET\_CONTROL은 한번에 여러개의 속성을 동시에 수정할 수 있습니다. ALIGNMENT 키워드는 좌측,가운데,우측 정렬을 결정할 수 있습니다. WIDGET\_TABLE의 키워드 설정값은 대부분, 단일값을 주면 모든 셀의 설정을 정의하게 되고, 배열을 주게 되면 각 셀의 설정값

을 따로 정의할 수 있게 됩니다. ALIGNMENT 키워드 역시 그렇습니다.

다음과 같이 설정하여 셀의 폭을 좁히고, 포맷을 소수점 아래 한자리까지만 하고, 가운데 정렬을 하도록 해보겠습니다.

```
widget_control, table, COLUMN_WIDTHS=40, $
FORMAT='(F5.1)', ALIGNMENT=1
```



셀 값을 수정할 수 있으므로 몇 개의 값을 변경해 보세요. 값이 변경되었으므로 처음 생성된 배열과는 다른 배열이 되어 있을 것입니다. 변경된 배열 값을 읽어 오는 것도 WIDGET\_CONTROL이 합니다.

```
widget_control, table, GET_VALUE=newdata
print, newdata ;결과를 확인해 보세요.
```

마지막으로 각 셀의 속성을 따로 지정하는 방법을 사용해 보겠습니다. 모든 셀의 정렬을 가운데로 하고, 4번 컬럼의 정렬만 우측으로 하고자 한다면 다음과 같이 할 수 있습니다.

```
align=replicate(1, 10, 10) ;일단 모든 셀 속성을 가운데 정렬
align[4, *] = 2 ;4번 컬럼은 우측 정렬
widget_control, table, ALIGNMENT=align ;속성 변경
```

## 셀의 배경색을 이용한 영상 표출

이상우박사의 블로그에서는 영상을 확대하여 디스플레이하고 확대된 픽셀의 중앙에 화소값을 써 넣는 방법을 소개하였습니다. 여기서는 WIDGET\_TABLE의 셀 색상을 조정하는 방법으로 같은 표출을 해 보겠습니다.

## 영상 읽기

```
file=filepath('moon_landing.png', subdir=['examples', 'data'])
data=read_png(file)
im=image(data, axis_style=1)
```



IDL의 영상 표출 방식은 좌측하단이 원점이 됩니다. 그림에서 보이는 좌표체계와 같이 수학좌표체계의 1사분면을 의미하는데, 테이블의 좌표체계는 우측 상단이 원점이 됩니다

다. 그러므로 영상을 상하 방향으로 뒤집어 주어야 IDL 영상표출 상태와 같아집니다.

```
data=reverse(data, 2) ;영상의 상하반전
sz=size(data, /dimension) ;영상의 크기 조사
```

### 화소값으로 GUI 테이블 생성

```
tlb=widget_base()
table=widget_table(tlb, value=data, SCR_XSIZE=800, $
SCR_YSIZE=800, ALIGNMENT=1, COLUMN_WIDTHS=20)
widget_control, tlb, /realize
```

### 글자색 바꾸기

```
widget_control, table, FOREGROUND_COLOR=[75, 0, 130]
```

WIDGET\_TABLE()의 색 체계는 [R, G, B] 조합을 사용합니다. 글자색으로 지정한 75(R), 0(G), 130(B) 조합은 'INDIGO' 라는 컬러를 의미합니다. 그래서 이러한 사용 방법도 유효합니다.

```
widget_control, table, FOREGROUND_COLOR=!color.INDIGO
```

### 컬러테이블 적용

WIDGET\_TABLE의 컬러체계는 R,G,B 조합(트루컬러)인데, 사용하는 영상은 0~255의 값을 가지는 인덱스 컬러 체계입니다. 흑백 컬러테이블(IDL의 컬러테이블 0번)을 이용하여 그림과 같이 흑백으로 보이는 것입니다. 그러므로 영상에 컬러테이블을 적용하여 트루컬러체계로 변환합니다.(한장 강의 문서 중, "인덱스 컬러와 RGB 트루컬러의 상호 전환"을 참고하세요).

```
loadct, 0
tvlct, r, g, b, /get
colors=bytarr(3, sz[0], sz[1])
colors[0, *, *]=r[data]
colors[1, *, *]=g[data]
colors[2, *, *]=b[data]
```


### 셀 배경색을 영상의 색으로 설정

```
widget_control,table, BACKGROUND_COLOR=colors
```

### 테이블의 위치(좌표) 이동

테이블에 있는 스크롤바를 이용해서 원하는 위치로 갈 수도 있지만, WIDGET\_CONTROL을 이용해서도 위치 이동이 가능합니다. 우주인의 머리 부분에 해당하는 부분으로 이동해 봅시다.

```
widget_control,table, SET_TABLE_VIEW=[145, 83]
```




### 다른 컬러테이블을 적용해 보기

컬러테이블을 적용하여 영상을 테이블로 표출하는 방법은 항상 같습니다. 5번 테이블을 이용하는 과정을 보세요.

```
loadct, 5
tvlct, r, g, b, /get
colors=bytarr(3, sz[0], sz[1])
colors[0, *, *]=r[data]
colors[1, *, *]=g[data]
colors[2, *, *]=b[data]
WIDGET_CONTROL, table, BACKGROUND_COLOR=colors
```

### 트루컬러 영상을 셀 배경으로 하는 테이블

```
file=filepath('rose.jpg', subdir=['examples', 'data'])
read_jpeg, file, rose
im=image(rose, axis_style=1)
tlb=widget_base()
strvalue=string(rose[0, *, *], format='(i3)'+','+$
string(rose[1, *, *], format='(i3)'+','+$
string(rose[2, *, *], format='(i3)')
table=widget_table(tlb, value=reform(strvalue, 227, 149), $
foreground_color=!color.ivory, background_color=rose, $
SCR_XSIZE=1000, SCR_YSIZE=800)
widget_control, tlb, /realize
widget_control, table, SET_TABLE_VIEW=[43, 60]
```



영상이 트루컬러라면 컬러테이블적용 단계는 필요가 없습니다. 바로 RGB 영상 변수 rose를 BACKGROUND\_COLOR로 지정하면 됩니다. 이 예제에서는 R, G, B 채널 값을 문자열로 연결하여 문자열 셀 값 배열을 만들었습니다. 이 예제에서 영상의 상하 반전을 하지는 않았습니다. (227, 149)는 영상의 크기입니다.